# inttypes.m4 serial 1 (gettext-0.11.4)
dnl Copyright (C) 1997-2002 Free Software Foundation, Inc.
dnl This file is free software; the Free Software Foundation
dnl gives unlimited permission to copy and/or distribute it,
dnl with or without modifications, as long as this notice is preserved.

dnl From Paul Eggert.

# Define HAVE_INTTYPES_H if <inttypes.h> exists and doesn't clash with
# <sys/types.h>.

AC_DEFUN([gt_HEADER_INTTYPES_H],
[
  if test "OS/390" = "`uname`"
  then
    gt_cv_header_inttypes_h=no
  else
    AC_CACHE_CHECK([for inttypes.h], gt_cv_header_inttypes_h,
    [
      AC_TRY_COMPILE(
        [#include <sys/types.h>
#include <inttypes.h>],
        [], gt_cv_header_inttypes_h=yes, gt_cv_header_inttypes_h=no)
    ])
    if test $gt_cv_header_inttypes_h = yes; then
      AC_DEFINE_UNQUOTED(HAVE_INTTYPES_H, 1,
        [Define if <inttypes.h> exists and doesn't clash with <sys/types.h>.])
    fi
  fi
])
